

by Curtis Cooper

The Fifteen Puzzle

The Fifteen Puzzle, a creation of puzzle wizard Sam Loyd, is a mathematical problem which can be simulated on a microcomputer. This game consists of 15 numbered squares and a space in a 4 x 4 tray. The puzzle starts with some permutation of the numbered tiles. The object of the Fifteen Puzzle is to rearrange the tiles by sliding them about the tray, until the numbered squares are in serial order. For example, the initial configuration of the Fifteen Puzzle may look like Figure 1 and the final, sought-after configuration is Figure 2.

1	2	3	4
5	6	7	8
9	10	11	12
13	14	15	

Figure 1

15	7	4	14
9	3	11	
13	8	10	6
5	2	12	1

Figure 2

Program Notes

The North Star BASIC program will let you enjoy this game and help improve your skill at solving it. It uses two arrays, an array ("A") dimensioned to 16 and an array T dimensioned to 16 x 5. The first array consists of the numbers 1 through 15 and the number 0, which represents the space. We can think of it as the puzzle board. With respect to the puzzle board in Figure 1, the array would appear as in Figure 3.

15	7	4	14	9	3	11	0	13	8	10	6	5	2	12	1
1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16

Figure 3 — Array "A".

Array T denotes the neighbors of a square on the puzzle board. For example, square 2 has 3 neighbors (square 1, square 3, and square 6) while square 7 has 4 neighbors (square 3, square 6, square 8, and square 11). T(i,1) represents the number of neighbors of square i, and T(i,2) through T(i,5) contains the list of those neighbors, filled-out with zeros if necessary. Thus, with respect to the finished puzzle (Figure 2), the two examples above would appear as

- | | |
|------------|--------------|
| T(2,1) = 2 | T(7,1) = 4 |
| T(2,2) = 1 | T(7,2) = 3 |
| T(2,3) = 3 | T(7,3) = 6 |
| T(2,4) = 6 | T(7,4) = 8 |
| T(2,5) = 0 | T(7,5) = 11. |

The program begins by explaining the Fifteen Puzzle. It asks for the difficulty level you wish to attempt and then randomly generates your scrambled puzzle. The program next prints out the arrangement of the numbers on the board and asks you to indicate the number of the tile you wish to move to the blank square. It moves your indicated tile to the blank space and checks to see if the tiles are in serial order. If they are, the number of moves you took to arrange the tiles in order is printed and you are asked if you want to try another puzzle. If they are not in serial order, you are asked to give another move, etc.

Figure 4 is a sample run of the program. The program listing begins on page 35.

References

- Gardner, Martin. *Scientific American*, Vol. 197 (Aug. 1957), 120.
- Gardner, Martin. *Scientific American*, Vol. 210 (Feb. 1964), 122.
- Liebeck, Hans. "Some Generalizations of the 14-15 Puzzle," *Mathematics Magazine*, Vol. 44 (1971), 185-189.

- Spitznagel, Edward L., Jr. "A New Look at the Fifteen Puzzle," *Mathematics Magazine*, Vol. 40 (1967) 171-174.

Figure 4.

SAMPLE RUN

THIS PROGRAM SIMULATES THE FAMOUS 'FIFTEEN PUZZLE'. THE OBJECT OF THE GAME IS TO START WITH A SCRAMBLED 4X4 ARRAY OF 15 NUMBERED TILES AND A SPACE AND ARRANGE THE TILES IN ORDER FROM 1 TO 15 WITH THE SPACE IN THE LOWER RIGHT CORNER.

WHAT LEVEL OF DIFFICULTY DO YOU WISH? (1,2,3,4) (4 IS HARDEST) ? 1

6	1	7	2
10	3	8	12
0	5	14	4
13	9	11	15

INDICATE THE NUMBER OF THE TILE YOU WISH TO MOVE TO THE BLANK SQUARE ? 10

6	1	7	2
0	3	8	12
10	5	14	4
13	9	11	15

INDICATE THE NUMBER OF THE TILE YOU WISH TO MOVE TO THE BLANK SQUARE ? 6

0	1	7	2
6	3	8	12
10	5	14	4
13	9	11	15

INDICATE THE NUMBER OF THE
TILE YOU WISH TO MOVE TO THE
BLANK SQUARE ? 1

1	2	3	4
5	6	7	8
9	10	11	12
13	14	0	15

INDICATE THE NUMBER OF THE
TILE YOU WISH TO MOVE TO THE
BLANK SQUARE ? 15

1	2	3	4
5	6	7	8
9	10	11	12
13	14	15	0

CONGRATULATIONS!!! YOU HAVE
FINALLY ARRANGED THE TILES
IN THE CORRECT ORDER. YOU
TOOK 156 MOVES.

DO YOU WISH TO TRY ANOTHER
POSSIBILITY ? (YES/NO) ? NO

GOODBYE. . . THANKS FOR PLAYING.

```

10 REM ***** THE FIFTEEN PUZZLE *****
20 REM ***** BY *****
30 REM ***** CURTIS N. COOPER *****
40 REM ***** SEPTEMBER 15, 1980 *****
50 DIM T(16,5),A(16)
60 FOR I = 1 TO 16
70 READ T(I,1),T(I,2),T(I,3),T(I,4),T(I,5)
80 NEXT I
90 PRINT "THIS PROGRAM SIMULATES THE FAMOUS"
100 PRINT " 'FIFTEEN PUZZLE'. THE OBJECT OF"
110 PRINT " 'THE GAME IS TO START WITH A"
120 PRINT " 'SCRAMBLED 4X4 ARRAY OF 15 NUMBERED"
130 PRINT " 'TILES AND A SPACE AND ARRANGE THE"
140 PRINT " 'TILES IN ORDER FROM 1 TO 15 WITH THE"
150 PRINT " 'SPACE IN THE LOWER RIGHT CORNER.'"
160 PRINT
170 J1=RND(-1)
180 PRINT
190 INPUT "WHAT LEVEL OF DIFFICULTY DO YOU WISH?
(1, 2, 3, 4) (4 IS HARDEST) ",C1
200 PRINT\PRINT\PRINT\PRINT\PRINT
210 REM THIS SECTION RANDOMLY GENERATES A FIFTEEN
PUZZLE OF *****
220 REM OF A PREDETERMINED DIFFICULTY *****
230 PRINT
240 C=0
250 FOR I = 1 TO 15
260 A(I) = I
270 NEXT I
280 A(16) = 0
290 K1 = 16
300 K0 = 0
310 J1 = RND(J1/100)
320 FOR I = 1 TO C1*50
330 J1 = RND(J1)
340 L = INT(T(K1,1)*J1+1)
350 K2 = T(K1,L+1)
360 IF K2=K0 THEN 330
370 K = A(K1)
380 A(K1) = A(K2)
390 A(K2) = K
400 K0 = K1
410 K1 = K2
420 NEXT I
430 REM *****
440 PRINT\PRINT\PRINT\PRINT\PRINT
450 GOSUB 990
460 PRINT\PRINT\PRINT\PRINT\PRINT
470 PRINT "INDICATE THE NUMBER OF THE TILE"
480 PRINT "YOU WISH TO MOVE TO THE BLANK SQUARE"
490 INPUT M
500 C = C + 1
510 PRINT CHR$(12)
520 FOR J = 1 TO 16
530 IF A(J) = M THEN EXIT 590
540 NEXT J
550 PRINT "INVALID INPUT, TRY AGAIN"
560 GOTO 440

```

continued on page 36.

The Next Whole Earth Catalog
Edited by Stewart Brand
Published by Random House
608 pages, \$12.50 paperbound
Reviewed by Julie Anton

First it was *The Whole Earth Catalog*, homely Bible of a generation who had dropped out and turned on in the 1960's, turned off a few years later, and split for the hills in search of something safe and homespun.

Whole Earth chronicler Stewart Brand watched his brainchild move from commune floors to chrome suburban coffee tables, hit the international best-sellers list, and win a controversial National Book Award. Troubled by fame, Brand officially called it quits with *The Last Whole Earth Catalog* in 1971. He managed one more *Epilogue* in 1974, but the Whole Earth flame was definitely out. He observed that "you cannot both be and defend a mountain range."

Now we have *The Next Whole Earth Catalog*. This is a 608-page distillation of the best elements of the earlier catalogs, with a vivid transformation for the 1980's, granting snap and modernity to the sense of practical pioneering and self-reliance which made its predecessors so important and unique.

"The image of me and mah woman and mah kids and mah dog and the chickens and the ducks and cows and the woods is a self-destructive fantasy after a while," Stewart Brand says, "because it eliminates a lot of quite wholesome dependencies from one's thinking." His latest catalog is not a Book of Exodus for a lost generation. We are not looking to escape from the 1980's. We are involved, and we like it that way.

Stewart Brand describes the purpose of his updated Whole Earth Catalog this way: "We are as gods and might as well get good at it. So far remotely done power and glory — as via government, big business, formal education, church — has succeeded to the point where gross defects obscure actual gains. In response to this dilemma and to these gains a realm of intimate, personal power is developing — the power of individuals to conduct their own education, find their own inspiration, shape their own environment, and share the adventure with whoever is interested. Tools that aid this process are sought and promoted by *The Next Whole Earth Catalog*."

There could be no greater symbol of this new Whole Earth philosophy than the computer, a tool which Stewart Brand expects will change our lives more

(continued on page 51)

The Fifteen Puzzle

Continued from page 35

```
570 REM THIS SECTION INTERCHANGES THE BLANK SPACE *****
580 REM AND THE NUMBERED TILE YOU MOVED *****
590 I = J
600 K = T(I,1)
610 FOR J = 1 TO K
620 L = T(I,J+1)
630 IF A(L) = 0 THEN EXIT 660
640 NEXT J
650 GOTO 550
660 K = T(I,J+1)
670 J = A(I)
680 A(I) = A(K)
690 A(K) = J
700 GOSUB 990
710 FOR J = 1 TO 15
720 IF A(J) <> J THEN EXIT 460
730 NEXT J
740 PRINT " CONGRATULATIONS!!! YOU HAVE FINALLY
ARRANGED THE TILES "
750 PRINT " IN THE CORRECT ORDER. YOU TOOK ",C," MOVES."
760 PRINT
770 INPUT " DO YOU WISH TO TRY ANOTHER POSSIBILITY ?
(YES/NO) ",Y$
780 IF Y$ = "YES" THEN 180
790 PRINT " GOODBYE. . . . . THANKS FOR PLAYING." END
800 REM THIS SECTION DEFINES THE NEIGHBORS OF EACH *****
810 REM SQUARE ON THE FIFTEEN PUZZLE BOARD *****
820 DATA 2,2,5,0,0
830 DATA 3,1,3,6,0
840 DATA 3,2,4,7,0
850 DATA 2,3,8,0,0
860 DATA 3,1,6,9,0
870 DATA 4,2,5,7,10
880 DATA 4,3,6,8,11
890 DATA 3,4,7,12,0
900 DATA 3,5,10,13,0
910 DATA 4,6,9,11,14
920 DATA 4,7,10,12,15
930 DATA 3,8,11,16,0
940 DATA 2,9,14,0,0
950 DATA 3,10,13,15,0
960 DATA 3,11,14,16,0
970 DATA 2,12,15,0,0
980 REM THIS SUBROUTINE PRINTS THE FIFTEEN PUZZLE BOARD *
990 PRINT %10I,A(1),%5I,A(2),%5I,A(3),%5I,A(4)
1000 PRINT
1010 PRINT %10I,A(5),%5I,A(6),%5I,A(7),%5I,A(8)
1020 PRINT
1030 PRINT %10I,A(9),%5I,A(10),%5I,A(11),%5I,A(12)
1040 PRINT
1050 PRINT %10I,A(13),%5I,A(14),%5I,A(15),%5I,A(16)
1060 PRINT
1070 RETURN
```